
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 7.1.2.11.2 which is part of the LTE/SAE test suite in the ‘iWD-TTCN3-B2015-03_D16wk10’ ATS delivery.

The test case can be demonstrated to run with one LTE-A UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
TC_7_1_2_11_2
Test Group:
LTE_A_R10_R11\7_1\MAC_CA.ttcn
ATS Version:
iwd-TTCN3-B2015-03_D16wk10
System Simulator used:
R&S CMW500


UE used:
Qualcomm MSM 8996
Verification Status:
PASS


4. Corrections required for TC 7.1.2.11.x
Change 1 – Corrections to TA configuration
	Sections / function name
	f_TC_7_1_2_11_Common ()

	Reason for change
	a)  As per TS 36.300 Annex J, a UE should cope with a maximum uplink transmission timing difference between TAGs of 32.47us for inter-band carrier aggregation with multiple TAGs.

b)   SCell needs to be switched on to meet the test purpose.
Core Spec references:

[TS 36.300 Annex J]

Carrier Aggregation

The reception timing difference at the physical layer of DL assignments and UL grants for the same TTI but from different serving cells (e.g. depending on number of control symbols, propagation and deployment scenario) does not affect MAC operation. A UE should cope with a relative propagation delay difference up to 30 us among the component carriers to be aggregated in both intra-band non-contiguous and inter-band non-contiguous CA. This implies that a UE should cope with a delay spread of up to 30.26 us among the component carriers monitored at the receiver, since the BS time alignment is specified to be up to 0.26 us. This also implies that the UE should cope with a maximum uplink transmission timing difference between TAGs of 32.47us for inter-band carrier aggregation with multiple TAGs.
Note: These changes require a Prose CR (see draft_R5-162xxx_TS-36523-1_TC-7-1-2-11-x).


	Summary of change
	a)   RAR TA values for SCell corrected for FDD and TDD accordingly.
b)   2nd Cell is switched on before SCell addition.

	TTCN module
	LTE_A_R10_R11\7_1\MAC_CA.ttcn

	MCC160 Comment
	See below


Change 2 – Corrections to PHR + SRS configuration
	Sections / function name
	· f_TC_7_1_2_11_Common ()
· f_EUTRA_508RRC_AddMod_1Scell ()

	Reason for change
	a)   Default PHR reporting interferes with the test purpose, i.e. with scheduled UL Data transmissions.

b)   Default Scell Deactivation timer expires during the execution of the TC body.
c)   SRS is not activated per default on the SCell, but it is required in oder to fulfil the test purpose.
d)   UE sends PHR report on activation of an SCell with configured uplink that is not handled in the scenario.
Core Spec references:

[TS 36.331 clause 6.3.2]

extendedPHR

Indicates if power headroom shall be reported using the Extended Power Headroom Report MAC control element defined in TS 36.321 [6] (value setup). E-UTRAN always configures the value setup if more than one Serving Cell with uplink is configured and if dual connectivity is not configured. E-UTRAN configures extendedPHR only if phr-Config is configured. The UE shall release extendedPHR if phr-Config is released.
[TS 36.321 clause 5.4.6]

Power Headroom Reporting

A Power Headroom Report (PHR) shall be triggered if any of the following events occur:

- activation of an SCell of any MAC entity with configured uplink.


	Summary of change
	a)   Since PHR reporting cannot be disabled for UL CA, the configuration should be so that PHR reports are not sent by the UE in this scenario.
b)   SCell Deactivation Timer is increased to not expire during the execution of the TC body.
c)   SCell addition procedure is extended to enable SRS configuration on the SCell.
d)   The scenario is modified to give an UL Grant for the PHR report after the SCell activation but before the Transparent Mode configuration to avoid MAC PDU forwarding to the TTCN.

	TTCN module
	· LTE_A_R10_R11\7_1\MAC_CA.ttcn
· Common\EUTRA_CA\EUTRA_CommonFunctions_CA.ttcn

	MCC160 Comment
	See below


Change 3 – Corrections to Cell ID parameter
	Sections / function name
	f_TC_7_1_2_11_Common ()

	Reason for change
	Incorrect Cell IDs used in various steps.

	Summary of change
	Cell IDs corrected.

	TTCN module
	LTE_A_R10_R11\7_1\MAC_CA.ttcn

	MCC160 Comment
	See below


Change 4 – Corrections to UL Grant config
	Sections / function name
	f_TC_7_1_2_11_Common ()

	Reason for change
	Function f_EUTRA_OneULGrantTransmission alway triggers to keep the PUCCH Synch automatically, that prohibits the SCell to get out of synch.

	Summary of change
	Triggers of certain UL grants are modified to fulfil the required behaviour of the TC body.

	TTCN module
	LTE_A_R10_R11\7_1\MAC_CA.ttcn

	MCC160 Comment
	See below


Change 5 – Corrections to Timing
	Sections / function name
	f_TC_7_1_2_11_Common ()

	Reason for change
	a)   The timing of certain steps is incorrect and not as per prose specification.
b)   The time span between Step 13 and Step 19 (consecutive grants on SCell) is too big using the current timing; which leads to expiry of the max possible SCell Deactivation Timer (rf128).

	Summary of change
	a)   Timing issues corrected according to the prose specification.

b)   Unnecessary timing is removed to avoid that SCell inactivity gets greater than the SCell Deactivation Timer.

	TTCN module
	LTE_A_R10_R11\7_1\MAC_CA.ttcn

	MCC160 Comment
	See below


Change 6 – Corrections to SRS re-initialization
	Sections / function name
	f_TC_7_1_2_11_Common ()

	Reason for change
	As per EUTRA MAC and RRC Core Specifications, SRS is not automatically restarted with the PDCCH order process after Step 17.

Core Spec references:

[TS 36.321 clause 5.2]

Maintenance of Uplink Time Alignment

- when a timeAlignmentTimer expires:

  - else if the timeAlignmentTimer is associated with an sTAG, then for all Serving Cells belonging to this TAG:

    - flush all HARQ buffers;

    - notify RRC to release SRS.
[TS 36.331 clause 5.3.13]

UE actions upon PUCCH/ SRS release request

Upon receiving an SRS release request from lower layers, the UE shall:

1> apply the default physical channel configuration for soundingRS-UL-ConfigDedicated, as specified in 9.2.4, for the cells of the concerned TAG.
[TS 36.331 clause 9.2.4]

Default physical channel configuration

SoundingRS-UL-ConfigDedicated
release.

Note: These changes require a Prose CR (see draft_R5-162xxx_TS-36523-1_TC-7-1-2-11-x).

	.Summary of change
	SRS is re-intialized on the SCell in new Step 17A.

	TTCN module
	LTE_A_R10_R11\7_1\MAC_CA.ttcn

	MCC160 Comment
	See below


Change 7 – Corrections to PDU #
	Sections / function name
	f_TC_7_1_2_11_Common ()

	Reason for change
	Wrong PDU # is sent in Step 20.

	Summary of change
	PDU # corrected in Step 20.

	TTCN module
	LTE_A_R10_R11\7_1\MAC_CA.ttcn

	MCC160 Comment
	See below


Change 8 – Corrections to Preamble Prefix
	Sections / function name
	f_TC_7_1_2_11_Common ()

	Reason for change
	Wrong / fixed PRACH Preamble Index is expected in Step 21, but there is no PDCCH order to assign this value.

	Summary of change
	Function is changed to accept any Index, since it is received in a common RACH procedure.

	TTCN module
	LTE_A_R10_R11\7_1\MAC_CA.ttcn

	MCC160 Comment
	See below


Change 9 – Corrections to CA type
	Sections / function name
	· f_TC_7_1_2_11_3 ()
· f_TC_7_1_2_10_3 ()     … similar C&P issue in an equivalent TC

	Reason for change
	Copy & Paste issue in the CA type for the non-contiguous CA combination.

	Summary of change
	CA type corrected.

	TTCN module
	LTE_A_R10_R11\7_1\MAC_CA.ttcn

	MCC160 Comment
	See below


·   Modifications in module ‘MAC_CA.ttcn’
	Function name
	f_TC_7_1_2_11_Common ()

	Reason for change
	See change 1.
See change 2.

See change 3.
See change 4.
See change 5.

See change 6.

See change 7.

See change 8.

See change 9.

	Summary of change
	See change 1.

See change 2.

See change 3.

See change 4.
See change 5.

See change 6.

See change 7.

See change 8.

See change 9.

	TTCN module
	LTE_A_R10_R11\7_1\MAC_CA.ttcn

	MCC160 Comment
	1. The requirements quoted are from stage 2 spec. Will wait for RAN5 to conclude.
2.Accepted
3.Accepted
4.Accepted in principle and implemented differently. Added new parameter ‘template (value)  PUCCH_Synch_Type p_PUCCH_Synch := cs_PUCCH_Synch_Auto_Def ‘ to f_EUTRA_OneULGrantTransmission and passed the parameter cs_PUCCH_Synch_None.
5.Accepted
6.Will Wait for RAN5 prose CR to agree.
7.Accepted
8. Accepted

9. Accepted


Before change
...
  /*
   * @desc      Common function : REFERENCE TS 36.523-1 clause 7.1.2.10.3
   * @status
   */
  function f_TC_7_1_2_10_3_EUTRA() runs on EUTRA_PTC
  { /*CA / Random access procedure / SCell / Intra-band Non Contiguous CA */
    f_TC_7_1_2_10_Common(eutra_Cell1, eutra_Cell3,CA_IntraBand); //@sic R5s150322 sic@
  }
  //----------------------------------------------------------------------------
  template (value) RachProcedureConfig_Type cs_RachProcedureConfig_Dedicated_71211(RACH_TimingAdvance_Type p_RAR_TA,
                                                                                   Dl_Bandwidth_Type p_Dl_Bandwidth) :=
  { /* Rach Response config for test case 7.1.2.10 */
    /* @status    */
    RACH_ConfigCommon := omit,
    RACH_ConfigDedicated := cs_RACH_ConfigDedicated_r8(cs_RACH_ConfigDedicated),
    RachProcedureList := {
      // preamble reception 1.  RAR Tx succesful and  No Contention Resolution as dedicated resource
      {
        RAResponse := cs_RandomAccessResponseConfig_Def(p_RAR_TA, p_Dl_Bandwidth,-,4),// RIV=0=> Nprb=1; Imcs=Itbs=4 => TB size =56
        ContentionResolutionCtrl := cs_ContReslCtrl_CRNTI_None // as dedicated resource indicated hence no PDCCH grant for UE needed.
      }
    }
  };
  /*
   * @desc      REFERENCE TS 36.523-1 clause 7.1.2.11
   */
  function f_TC_7_1_2_11_Common(EUTRA_CellId_Type p_PCellId,
                                EUTRA_CellId_Type p_SCellId,
                                CA_Tested_Type  p_CA_Type) runs on EUTRA_PTC
  {/* L2/MAC
    * 7.1.2.11 : CA / Maintenance of uplink time alignment / Multiple TA
    */
    var boolean v_UL_CA := true;
//    var boolean v_SendStatusPdu := false;
    var SubFrameTiming_Type v_SubFrameInfo, v_SubFrameInfo2;
    var SubFrameTiming_Type v_SubFrameTimingRX;
    var MAC_SDU_Type v_EncodedRlcPdu1;
    var MAC_SDU_Type v_EncodedRlcPdu2;
    var MAC_SDU_Type v_EncodedRlcPdu3;
    var octetstring v_EncodedPdcpPdu1, v_EncodedPdcpPdu2, v_EncodedPdcpPdu3;
    var RACH_TimingAdvance_Type v_RAR_TA;
    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;
    var SYSTEM_IND     v_SYSTEM_IND;
    var SubFrameTiming_Type v_Timing1;
    var SYSTEM_IND v_ReceivedSYSIND;
    var default v_DefaultVar := null;
    var boolean v_SetVerdictFalse:=false;
    ///////////////////////////////////////////////////////////////////////////////////
    //      PREAMBLE
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_Init(c3);
    f_EUTRA_CA_InitCells({p_PCellId, p_SCellId}, p_CA_Type, c3);
    //Get cell dependant parameters
    //Set C-RNTI to be used in s-cell creation
    f_EUTRA_CellInfo_SetC_RNTI(p_SCellId, tsc_C_RNTI_Def);
    //Create and configure all cells
    f_EUTRA_CellConfig_Pcell_Def(p_PCellId);
    f_EUTRA_CellConfig_Scell(p_SCellId, Scell_Active); //No DRB's created
    // bring UE to initial state
    f_EUTRA_Preamble(p_PCellId, STATE2A_TESTLOOP_ModeA);
    // creates cell and performs registration
    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm
    // Go to state 4
    f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling_DSR(p_PCellId, n8);
    //@sic R5-101183 sic@
    f_Delay(0.660); // 600 ms previous time alignment timer
    ///////////////////////////////////////////////////////////////////////////////////
    //      Test Body
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_TestBody_Set(true, layer2);
    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(p_PCellId);
    if (v_EUTRA_FDD_TDD_Mode == FDD) {
      v_RAR_TA := 600;
    } else {
      v_RAR_TA := 160;
    }
    //@siclog "Step 1-2" siclog@
    //Transmit RRCConnectionReconfiguration
    //Receive RRCConnectionReconfigurationComplete
    f_EUTRA_508RRC_AddMod_1Scell(p_PCellId,
                                 p_SCellId,
                                 1, //SCellIndex_r10
                                 p_CA_Type, // CA Type
                                 v_UL_CA,// UL CA desired
                                 -, //C-RNTI
                                 -, // p_MeasGapConfig
                                 -, //p_RRCConnectionReconfiguration
                                 -, //p_DciUlInfo_SCell
                                 -, //p_UL_GrantConfig_SCell
                                 -, //p_STAG_ToReleaseList
                                 {cs_STAG_ToAddMod_Def(tsc_STAG_ID1)},// p_STAG_ToAddModList
                                 cs_MAC_MainConfigSCell_Def(tsc_STAG_ID1),//p_MAC_MainConfigSCell_r11
                                 cs_RACH_ConfigCommonSCell_r11_Def,
                                 cs_PRACH_Config_Def(f_EUTRA_CellInfo_GetPrach_Config(p_SCellId)));
    f_SS_Config1TM_DRB_NoMACHeader(p_PCellId, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDLOnly);
    f_EUTRA_StartULGrantTransmissionOnSR_Reception(p_PCellId, cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def);
    //@siclog "Step 3" siclog@
    // Send Scell Activation Command
    f_EUTRA_ScellActivate_Deactivate(p_PCellId, tsc_Scell1_Activate);
    // Configure Scell for RACH procedure
    f_EUTRA_SS_ConfigRachProcedure(p_SCellId, tsc_C_RNTI_Def, cs_RachProcedureConfig_Dedicated_71211(f_EUTRA_CellInfo_GetRAR_TA(p_SCellId),
                                                                                                    f_EUTRA_CellInfo_GetDL_ChBandwidth(p_SCellId) ));
    // Confiure Pcell for RACH procedure
    f_EUTRA_SS_ConfigRachProcedure (eutra_Cell1, omit, cs_RachProcedureConfig_CRNTI(f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1),
                                                                                    f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1)) );
    // config for PReamble reception indication in Scell
    f_EUTRA_SS_ConfigRachPreambleIndMode(p_SCellId, tsc_L1Mac_IndicationMode_Enable);
    // config for PReamble reception indication in Pcell
    f_EUTRA_SS_ConfigRachPreambleIndMode(p_PCellId, tsc_L1Mac_IndicationMode_Enable);
    // Precode RLC PDU's
    v_EncodedPdcpPdu1 :=f_PDCP_PDU_Encvalue(cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(int2bit(0,12), crs_PDCP_SDU_36B)));
    v_EncodedRlcPdu1 := f_RLC_PDU_Encvalue(cs_RLC_AMD_1PDU(tsc_NoPoll,
                                                           int2bit(0,10),
                                                           v_EncodedPdcpPdu1 ));
    v_EncodedPdcpPdu2 := f_PDCP_PDU_Encvalue(cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(int2bit(1,12), crs_PDCP_SDU_36B1)));
    v_EncodedRlcPdu2 := f_RLC_PDU_Encvalue(cs_RLC_AMD_1PDU(tsc_NoPoll,
                                                           int2bit(1,10),
                                                           v_EncodedPdcpPdu2 ));
    v_EncodedPdcpPdu3 := f_PDCP_PDU_Encvalue(cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(int2bit(2,12), crs_PDCP_SDU_36B2)));
    v_EncodedRlcPdu3 := f_RLC_PDU_Encvalue(cs_RLC_AMD_1PDU(tsc_NoPoll,
                                                           int2bit(2,10),
                                                           v_EncodedPdcpPdu3 ));
    // Evaluate a time in future
     v_SubFrameInfo := f_EUTRA_GetNextSendOccasion(p_PCellId, 200); // 200 ms in future
    //@siclog "Step 4" siclog@
    // Send DL PDCCH order on CC2
    f_Send_PDCCH_Order(p_SCellId, cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number), cs_RA_PDCCH_Order(tsc_EUTRA_PrachPreambleIndex_Dedicated_Def));
    //@siclog "Step 5" siclog@
    SYSIND.receive ( car_PRACH_Preamble_IND (p_SCellId, cr_TimingInfo_Any));
    //@siclog "Step 6" siclog@
    // Taken care by SS due to preconfiguration
    v_SubFrameInfo :=  f_EUTRA_GetNextSendOccasion(p_PCellId, 100); // 100 ms in future
    //@siclog "Step 7" siclog@
    // Send DL PDU on CC1
    DRB.send(cas_DRB_COMMOM_REQ_MAC_PDU(p_SCellId,
                                        tsc_RbId_DRB1,
                                        cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number),
                                        cs_MAC_PDUList_1(cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRlcPdu1))));
   //@siclog "Step 8" siclog@
   //Send an uplink grant
   v_SubFrameInfo := f_EUTRA_TimingInfoAdd(v_SubFrameInfo, 60); // 60 ms later
   f_EUTRA_OneULGrantTransmission( p_SCellId,
                                    cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number),
                                    7, 3);  //results in a grant of 328 bits= 41 B
    //@siclog "Step 9" siclog@
    // Receive loop back PDU on CC2
    DRB.receive (car_DRB_COMMON_IND_RLC_PDUList (p_SCellId,
                                                 tsc_RbId_DRB1,
                                                 cr_TimingInfo_Any,
                                                 {cr_RLC_AMD_1PDU(?, int2bit(0, 10), v_EncodedPdcpPdu1)}));
   f_RLC_Status_PDU_Tx(tsc_LchId_DRB1,1,p_SCellId);
     //@siclog "Step 10" siclog@
     v_SubFrameInfo := f_EUTRA_GetNextSendOccasion(p_PCellId, 200); // 200 ms in future
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU(p_SCellId,tsc_RbId_DRB1, cs_TimingInfo_Now, cs_MAC_PDUList_1 (cs_MAC_PDU_TimingAdvance_0_STAG1)));
     v_SubFrameInfo2 := f_EUTRA_TimingInfoAdd(v_SubFrameInfo, 100); // 100 ms later
    // Configure SS for SRS reporting
    f_SS_ConfigSRS_IndMode (p_SCellId, tsc_L1Mac_IndicationMode_Enable,cs_TimingInfo(v_SubFrameInfo2.SFN.Number, v_SubFrameInfo2.Subframe.Number) );
    //@siclog "Step 11" siclog@
    // Send DL PDU on CC1
    DRB.send(cas_DRB_COMMOM_REQ_MAC_PDU(p_SCellId,
                                        tsc_RbId_DRB1,
                                        cs_TimingInfo(v_SubFrameInfo2.SFN.Number, v_SubFrameInfo2.Subframe.Number),
                                        cs_MAC_PDUList_1(cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRlcPdu2))));
   //@siclog "Step 12" siclog@
   //Send an uplink grant
   v_SubFrameInfo2 := f_EUTRA_TimingInfoAdd(v_SubFrameInfo2, 60); // 60 ms later
   f_EUTRA_OneULGrantTransmission( p_SCellId,
                                    cs_TimingInfo(v_SubFrameInfo2.SFN.Number, v_SubFrameInfo2.Subframe.Number),
                                    7, 3);  //results in a grant of 328 bits= 41 B
    //@siclog "Step 13" siclog@
    // Receive loop back PDU on CC2
    DRB.receive (car_DRB_COMMON_IND_RLC_PDUList (p_SCellId,
                                                 tsc_RbId_DRB1,
                                                 cr_TimingInfo_Any,
                                                 {cr_RLC_AMD_1PDU(?, int2bit(1, 10), v_EncodedPdcpPdu2)}));
    f_RLC_Status_PDU_Tx(tsc_LchId_DRB1,2,p_SCellId);
    //@siclog "Step 14" siclog@
    SYSIND.receive(car_SRS_IND(p_SCellId, cr_TimingInfo_Any, tsc_SRS_Stop)) -> value v_ReceivedSYSIND;
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");
    v_Timing1 := v_ReceivedSYSIND.Common.TimingInfo.SubFrame;
    if (f_EUTRA_SubFrameTimingCheckDuration(eutra_Cell1, l2Timer, f_EUTRA_SubFrameTimingDuration(v_SubFrameInfo, v_Timing1), 750)) {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 14: UE stopped periodic SRS ");
    }
    else {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14: UE stopped period srs  but wrong time");
    }
    v_DefaultVar := activate(a_EUTRA_MAC_SRSInd(p_SCellId,tsc_SRS_Start ));
    //Check UE stops SRS
    //Check time difference between step 10 and 14 to be Time alignment timer + tolerance.
    v_SubFrameInfo := f_EUTRA_GetNextSendOccasion(p_PCellId, 200); // 200 ms in future
    //@siclog "Step 15" siclog@
    // Send DL PDCCH order on CC2
    f_Send_PDCCH_Order(p_SCellId, cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number), cs_RA_PDCCH_Order(tsc_EUTRA_PrachPreambleIndex_Dedicated_Def));
    //@siclog "Step 16" siclog@
    SYSIND.receive ( car_PRACH_Preamble_IND (p_SCellId, cr_TimingInfo_Any));
    //@siclog "Step 17" siclog@
    // Taken care by SS due to preconfiguration
    v_SubFrameInfo :=  f_EUTRA_GetNextSendOccasion(p_PCellId, 100); // 100 ms in future
    f_EUTRA_StopULGrantTransmission_No_PUCCH_Sync (p_PCellId,cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number) );
    //@siclog "Step 18" siclog@
    v_SubFrameInfo :=  f_EUTRA_TimingInfoAdd(v_SubFrameInfo, 100); // 100 ms in future
    //Send TA for PTAG
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU(p_PCellId,tsc_RbId_DRB1, cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number), cs_MAC_PDUList_1 (cs_MAC_PDU_TimingAdvance_0)));
    //@siclog "Step 19" siclog@
    // Send TA for Stag1
    v_SubFrameInfo2 :=  f_EUTRA_TimingInfoAdd(v_SubFrameInfo, 300); // 100 ms in future
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU(p_SCellId,tsc_RbId_DRB1, cs_TimingInfo(v_SubFrameInfo2.SFN.Number, v_SubFrameInfo2.Subframe.Number), cs_MAC_PDUList_1 (cs_MAC_PDU_TimingAdvance_0_STAG1)));
    //@siclog "Step 20" siclog@
    // Send DL PDU on CC1
    v_SubFrameInfo2 :=  f_EUTRA_TimingInfoAdd(v_SubFrameInfo, 300); // 100 ms in future
    DRB.send(cas_DRB_COMMOM_REQ_MAC_PDU(p_SCellId,
                                        tsc_RbId_DRB1,
                                        cs_TimingInfo(v_SubFrameInfo2.SFN.Number, v_SubFrameInfo2.Subframe.Number),
                                        cs_MAC_PDUList_1(cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRlcPdu1))));
    //Check UE stops SRS in parallel to step 22
   //@siclog "Step 21-22" siclog@
   interleave {
       []SYSIND.receive ( car_PRACH_Preamble_IND (p_PCellId, cr_TimingInfo_Any)) -> value v_SYSTEM_IND
       {
            v_SubFrameTimingRX := v_SYSTEM_IND.Common.TimingInfo.SubFrame;
             if (f_EUTRA_SubFrameTimingDuration_Float(v_SubFrameInfo2, v_SubFrameTimingRX) <= 0.225)
               {
                     f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 22");
                  }
             else
                 {
                       //f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 22");
                       v_SetVerdictFalse := true;
                 }
       }
     [] SYSIND.receive(car_SRS_IND(p_SCellId, cr_TimingInfo_Any, tsc_SRS_Stop)) {}
   };
   // defined to avoid usage of communication operation not allowed inside an interleave statement
   if (v_SetVerdictFalse) {
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 22");
   }
   //@siclog "Step 23" siclog@
   //Send an uplink grant
   f_EUTRA_OneULGrantTransmission( p_PCellId,
                                    cs_TimingInfo_Now,
                                    7, 3);  //results in a grant of 328 bits= 41 B
   //@siclog "Step 24" siclog@
   DRB.receive (car_DRB_COMMON_IND_RLC_PDUList (p_PCellId,
                                                 tsc_RbId_DRB1,
                                                 cr_TimingInfo_Any,
                                                 {cr_RLC_AMD_1PDU(?, int2bit(2, 10), v_EncodedPdcpPdu3)}));
    // config for stop of PReamble reception indication in Scell
    f_EUTRA_SS_ConfigRachPreambleIndMode(p_SCellId, tsc_L1Mac_IndicationMode_Disable);
    // config for stop of PReamble reception indication in Pcell
    f_EUTRA_SS_ConfigRachPreambleIndMode(p_PCellId, tsc_L1Mac_IndicationMode_Disable);
    // Config for stop of SRS toggle indication
    f_SS_ConfigSRS_IndMode (p_SCellId, tsc_L1Mac_IndicationMode_Disable );
    // deactivate reception of SRS's in alt step
    deactivate(v_DefaultVar);
    //@siclog "Step 25" siclog@
    //Send RLC status PDU on CC2
    f_RLC_Status_PDU_Tx(tsc_LchId_DRB1,3,p_PCellId);
    ///////////////////////
    // Postamble
    ///////////////////////
    f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(p_PCellId, cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def);
    f_EUTRA_TestBody_Set(false);
    //Switch/power off UE
    f_EUTRA_Postamble(p_PCellId, E2_CONNECTED);
  }
  /*
   * @desc      Common function : REFERENCE TS 36.523-1 clause 7.1.2.11.1
   * @status
   */
  function f_TC_7_1_2_11_1_EUTRA() runs on EUTRA_PTC
  { /*CA / Maintenance of uplink time alignment / Multiple TA/ Intra-band Contiguous CA */
    f_TC_7_1_2_11_Common(eutra_Cell1, eutra_Cell3,CA_IntraBand);
  }
  /*
   * @desc      Common function : REFERENCE TS 36.523-1 clause 7.1.2.11.2
   * @status
   */
  function f_TC_7_1_2_11_2_EUTRA() runs on EUTRA_PTC
  { /* CA / Maintenance of uplink time alignment / Multiple TA /  Inter-Band CA */
    f_TC_7_1_2_11_Common(eutra_Cell1, eutra_Cell10, CA_InterBand);
  }
  /*
   * @desc      Common function : REFERENCE TS 36.523-1 clause 7.1.2.11.3
   * @status
   */
  function f_TC_7_1_2_11_3_EUTRA() runs on EUTRA_PTC
  { /*CA / Maintenance of uplink time alignment / Multiple TA/ Intra-band Non Contiguous CA */
    f_TC_7_1_2_11_Common(eutra_Cell1, eutra_Cell3,CA_IntraBand);//@sic R5s150322 sic@
  }
}
...

After change
...
  /*
   * @desc      Common function : REFERENCE TS 36.523-1 clause 7.1.2.10.3
   * @status
   */
  function f_TC_7_1_2_10_3_EUTRA() runs on EUTRA_PTC
  { /*CA / Random access procedure / SCell / Intra-band Non Contiguous CA */
    f_TC_7_1_2_10_Common(eutra_Cell1, eutra_Cell3, CA_IntraBandNonContiguous); //@sic R5s150322 sic@
  }
  //----------------------------------------------------------------------------
  template (value) RachProcedureConfig_Type cs_RachProcedureConfig_Dedicated_71211(RACH_TimingAdvance_Type p_RAR_TA,
                                                                                   Dl_Bandwidth_Type p_Dl_Bandwidth) :=
  { /* Rach Response config for test case 7.1.2.10 */
    /* @status    */
    RACH_ConfigCommon := omit,
    RACH_ConfigDedicated := cs_RACH_ConfigDedicated_r8(cs_RACH_ConfigDedicated),
    RachProcedureList := {
      // preamble reception 1.  RAR Tx succesful and  No Contention Resolution as dedicated resource
      {
        RAResponse := cs_RandomAccessResponseConfig_Def(p_RAR_TA, p_Dl_Bandwidth,-,4),// RIV=0=> Nprb=1; Imcs=Itbs=4 => TB size =56
        ContentionResolutionCtrl := cs_ContReslCtrl_CRNTI_None // as dedicated resource indicated hence no PDCCH grant for UE needed.
      }
    }
  };
  /*
   * @desc      REFERENCE TS 36.523-1 clause 7.1.2.11
   */
  function f_TC_7_1_2_11_Common(EUTRA_CellId_Type p_PCellId,
                                EUTRA_CellId_Type p_SCellId,
                                CA_Tested_Type  p_CA_Type) runs on EUTRA_PTC
  {/* L2/MAC
    * 7.1.2.11 : CA / Maintenance of uplink time alignment / Multiple TA
    */
    var boolean v_UL_CA := true;
//    var boolean v_SendStatusPdu := false;
    var SubFrameTiming_Type v_SubFrameInfo, v_SubFrameInfo2;
    var SubFrameTiming_Type v_SubFrameTimingRX;
    var MAC_SDU_Type v_EncodedRlcPdu1;
    var MAC_SDU_Type v_EncodedRlcPdu2;
    var MAC_SDU_Type v_EncodedRlcPdu3;
    var octetstring v_EncodedPdcpPdu1, v_EncodedPdcpPdu2, v_EncodedPdcpPdu3;
    var RACH_TimingAdvance_Type v_RAR_TA;
    var EUTRA_FDD_TDD_Mode_Type v_EUTRA_FDD_TDD_Mode;
    var SYSTEM_IND     v_SYSTEM_IND;
    var SubFrameTiming_Type v_Timing1;
    var SYSTEM_IND v_ReceivedSYSIND;
    var default v_DefaultVar := null;
    var boolean v_SetVerdictFalse:=false;
    var EUTRA_FDD_TDD_CellInfo_Type v_FDD_TDDInfo;
    var template (value) SoundingRS_UL_ConfigDedicated v_SoundingRS_UL_ConfigDedicated;
    var template (omit)  DedicatedSCell_UL_Configuration_Type v_DedicatedSCell_UL_Configuration;
    var template (value) DL_DCCH_Message v_RRCConnectionReconfiguration;
    ///////////////////////////////////////////////////////////////////////////////////
    //      PREAMBLE
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_Init(c3);
    f_EUTRA_CA_InitCells({p_PCellId, p_SCellId}, p_CA_Type, c3);
    //Get cell dependant parameters
    //Set C-RNTI to be used in s-cell creation
    f_EUTRA_CellInfo_SetC_RNTI(p_SCellId, tsc_C_RNTI_Def);
    //Create and configure all cells
    f_EUTRA_CellConfig_Pcell_Def(p_PCellId);
    f_EUTRA_CellConfig_Scell(p_SCellId, Scell_Active); //No DRB's created
    // bring UE to initial state
    f_EUTRA_Preamble(p_PCellId, STATE2A_TESTLOOP_ModeA);
    // creates cell and performs registration
    f_EUTRA_AS_CipheringAlgorithm_Set(eea0); // set null AS ciphering algorithm
    // Go to state 4
    f_EUTRA_LoopBackActivation_State4_TA_Infinity_LB_NoScaling_DSR(p_PCellId, n8);
    //@sic R5-101183 sic@
    f_Delay(0.660); // 600 ms previous time alignment timer
    ///////////////////////////////////////////////////////////////////////////////////
    //      Test Body
    //////////////////////////////////////////////////////////////////////////////////
    f_EUTRA_TestBody_Set(true, layer2);
    v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD(p_PCellId);
    if (v_EUTRA_FDD_TDD_Mode == FDD) {
      v_RAR_TA := 61;
    } else {
      v_RAR_TA := 22;
    }
    f_EUTRA_SetCellPower (p_SCellId, tsc_ServingCellRS_EPRE);
    //@siclog "Step 1-2" siclog@
    //Transmit RRCConnectionReconfiguration
    //Receive RRCConnectionReconfigurationComplete
    f_EUTRA_508RRC_AddMod_1Scell(p_PCellId,
                                 p_SCellId,
                                 1, //SCellIndex_r10
                                 p_CA_Type, // CA Type
                                 v_UL_CA,// UL CA desired
                                 -, //C-RNTI
                                 -, // p_MeasGapConfig
                                 -, //p_RRCConnectionReconfiguration
                                 -, //p_DciUlInfo_SCell
                                 -, //p_UL_GrantConfig_SCell
                                 -, //p_STAG_ToReleaseList
                                 {cs_STAG_ToAddMod_Def(tsc_STAG_ID1)},// p_STAG_ToAddModList
                                 cs_MAC_MainConfigSCell_Def(tsc_STAG_ID1),//p_MAC_MainConfigSCell_r11
                                 cs_RACH_ConfigCommonSCell_r11_Def,
                                 cs_PRACH_Config_Def(f_EUTRA_CellInfo_GetPrach_Config(p_SCellId)),
                                 -,
                                 rf128,
                                 -,
                                 cs_PHR_Config_Setup(infinity_, sf1000, infinity_),
                                 true);
    //@siclog "Step 3" siclog@
    // Send Scell Activation Command
    f_EUTRA_SS_CommonActivateScellConfig(p_PCellId, cas_ScellActivate_REQ(p_PCellId, tsc_Scell1_Activate, true));
    //Handle ePHR report on activation of SCell with configured uplink
    f_EUTRA_OneULGrantTransmission( p_PCellId,
                                    cs_TimingInfo_Now,
                                    9, 25 );
    f_SS_Config1TM_DRB_NoMACHeader(p_PCellId, tsc_RbId_DRB1, cs_MAC_TestMode_NoHeaderManipulationDLOnly);
    // Configure Scell for RACH procedure
    f_EUTRA_SS_ConfigRachProcedure(p_SCellId, tsc_C_RNTI_Def, cs_RachProcedureConfig_Dedicated_71211(v_RAR_TA,
                                                                                                    f_EUTRA_CellInfo_GetDL_ChBandwidth(p_SCellId) ));
    // Configure Pcell for RACH procedure
    f_EUTRA_SS_ConfigRachProcedure (p_PCellId, omit, cs_RachProcedureConfig_CRNTI(f_EUTRA_CellInfo_GetRAR_TA(p_PCellId),
                                                                                    f_EUTRA_CellInfo_GetDL_ChBandwidth(p_PCellId)) );
    // config for PReamble reception indication in Scell
    f_EUTRA_SS_ConfigRachPreambleIndMode(p_SCellId, tsc_L1Mac_IndicationMode_Enable);
    // config for PReamble reception indication in Pcell
    f_EUTRA_SS_ConfigRachPreambleIndMode(p_PCellId, tsc_L1Mac_IndicationMode_Enable);
    // Precode RLC PDU's
    v_EncodedPdcpPdu1 :=f_PDCP_PDU_Encvalue(cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(int2bit(0,12), crs_PDCP_SDU_36B)));
    v_EncodedRlcPdu1 := f_RLC_PDU_Encvalue(cs_RLC_AMD_1PDU(tsc_NoPoll,
                                                           int2bit(0,10),
                                                           v_EncodedPdcpPdu1 ));
    v_EncodedPdcpPdu2 := f_PDCP_PDU_Encvalue(cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(int2bit(1,12), crs_PDCP_SDU_36B1)));
    v_EncodedRlcPdu2 := f_RLC_PDU_Encvalue(cs_RLC_AMD_1PDU(tsc_NoPoll,
                                                           int2bit(1,10),
                                                           v_EncodedPdcpPdu2 ));
    v_EncodedPdcpPdu3 := f_PDCP_PDU_Encvalue(cs_PDCP_PDU_LongSN(cs_PDCP_DataPdu_LongSN(int2bit(2,12), crs_PDCP_SDU_36B2)));
    v_EncodedRlcPdu3 := f_RLC_PDU_Encvalue(cs_RLC_AMD_1PDU(tsc_NoPoll,
                                                           int2bit(2,10),
                                                           v_EncodedPdcpPdu3 ));
    // Evaluate a time in future
     v_SubFrameInfo := f_EUTRA_GetNextSendOccasion(p_PCellId, 200); // 200 ms in future
    //@siclog "Step 4" siclog@
    // Send DL PDCCH order on CC2
    f_Send_PDCCH_Order(p_SCellId, cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number), cs_RA_PDCCH_Order(tsc_EUTRA_PrachPreambleIndex_Dedicated_Def));
    //@siclog "Step 5" siclog@
    SYSIND.receive ( car_PRACH_Preamble_IND (p_SCellId, cr_TimingInfo_Any));
    //@siclog "Step 6" siclog@
    // Taken care by SS due to preconfiguration
    v_SubFrameInfo :=  f_EUTRA_GetNextSendOccasion(p_PCellId, 100); // 100 ms in future
    //@siclog "Step 7" siclog@
    // Send DL PDU on CC1
    DRB.send(cas_DRB_COMMOM_REQ_MAC_PDU(p_SCellId,
                                        tsc_RbId_DRB1,
                                        cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number),
                                        cs_MAC_PDUList_1(cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRlcPdu1))));
   //@siclog "Step 8" siclog@
   //Send an uplink grant
   v_SubFrameInfo := f_EUTRA_TimingInfoAdd(v_SubFrameInfo, 60); // 60 ms later
   f_EUTRA_ULGrantConfiguration_Common( p_SCellId,
                                        cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number),
                                        cs_DciInfo_CcchDcchDtchUL_Explicit(7, 3, cs_TransRetransmissionListUL_Def),  //results in a grant of 328 bits= 41 B
                                        cs_PUCCH_Synch_None,
                                        cs_UL_GrantConfig_OneTime);
    //@siclog "Step 9" siclog@
    // Receive loop back PDU on CC2
    DRB.receive (car_DRB_COMMON_IND_RLC_PDUList (p_SCellId,
                                                 tsc_RbId_DRB1,
                                                 cr_TimingInfo_Any,
                                                 {cr_RLC_AMD_1PDU(?, int2bit(0, 10), v_EncodedPdcpPdu1)}));
   f_RLC_Status_PDU_Tx(tsc_LchId_DRB1,1,p_SCellId);
     //@siclog "Step 10" siclog@
     v_SubFrameInfo := f_EUTRA_GetNextSendOccasion(p_PCellId, 200); // 200 ms in future
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU(p_SCellId,tsc_RbId_DRB1, cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number), cs_MAC_PDUList_1 (cs_MAC_PDU_TimingAdvance_0_STAG1)));
     v_SubFrameInfo2 := f_EUTRA_TimingInfoAdd(v_SubFrameInfo, 100); // 100 ms later
    // Configure SS for SRS reporting
    f_SS_ConfigSRS_IndMode (p_SCellId, tsc_L1Mac_IndicationMode_Enable,cs_TimingInfo(v_SubFrameInfo2.SFN.Number, v_SubFrameInfo2.Subframe.Number) );
    //@siclog "Step 11" siclog@
    // Send DL PDU on CC1
    DRB.send(cas_DRB_COMMOM_REQ_MAC_PDU(p_SCellId,
                                        tsc_RbId_DRB1,
                                        cs_TimingInfo(v_SubFrameInfo2.SFN.Number, v_SubFrameInfo2.Subframe.Number),
                                        cs_MAC_PDUList_1(cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRlcPdu2))));
   //@siclog "Step 12" siclog@
   //Send an uplink grant
   v_SubFrameInfo2 := f_EUTRA_TimingInfoAdd(v_SubFrameInfo2, 60); // 60 ms later
   f_EUTRA_ULGrantConfiguration_Common( p_SCellId,
                                        cs_TimingInfo(v_SubFrameInfo2.SFN.Number, v_SubFrameInfo2.Subframe.Number),
                                        cs_DciInfo_CcchDcchDtchUL_Explicit(7, 3, cs_TransRetransmissionListUL_Def),  //results in a grant of 328 bits= 41 B
                                        cs_PUCCH_Synch_KeepAsItIs,
                                        cs_UL_GrantConfig_OneTime);
    //@siclog "Step 13" siclog@
    // Receive loop back PDU on CC2
    DRB.receive (car_DRB_COMMON_IND_RLC_PDUList (p_SCellId,
                                                 tsc_RbId_DRB1,
                                                 cr_TimingInfo_Any,
                                                 {cr_RLC_AMD_1PDU(?, int2bit(1, 10), v_EncodedPdcpPdu2)}));
    f_RLC_Status_PDU_Tx(tsc_LchId_DRB1,2,p_SCellId);
    //@siclog "Step 14" siclog@
    SYSIND.receive(car_SRS_IND(p_SCellId, cr_TimingInfo_Any, tsc_SRS_Stop)) -> value v_ReceivedSYSIND;
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");
    v_Timing1 := v_ReceivedSYSIND.Common.TimingInfo.SubFrame;
    if (f_EUTRA_SubFrameTimingCheckDuration(p_PCellId, l2Timer, f_EUTRA_SubFrameTimingDuration(v_SubFrameInfo, v_Timing1), 750)) {
      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 14: UE stopped periodic SRS ");
    }
    else {
      f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14: UE stopped period srs  but wrong time");
    }
    v_DefaultVar := activate(a_EUTRA_MAC_SRSInd(p_SCellId,tsc_SRS_Start ));
    //Check UE stops SRS
    //Check time difference between step 10 and 14 to be Time alignment timer + tolerance.
    //@siclog "Step 15" siclog@
    // Send DL PDCCH order on CC2
    f_Send_PDCCH_Order(p_SCellId, cs_TimingInfo_Now, cs_RA_PDCCH_Order(tsc_EUTRA_PrachPreambleIndex_Dedicated_Def));
    //@siclog "Step 16" siclog@
    SYSIND.receive ( car_PRACH_Preamble_IND (p_SCellId, cr_TimingInfo_Any));
    //@siclog "Step 17" siclog@
    // Taken care by SS due to preconfiguration
    //@siclog "Step 17A" siclog@
    //Start SRS on SCell
    f_EUTRA_StartULGrantTransmissionOnSR_Reception(p_PCellId, cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def);
    v_FDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info (p_SCellId);
    if (v_FDD_TDDInfo.FDD_TDD == TDD){
      v_SoundingRS_UL_ConfigDedicated := cds_508_SoundingRsUl_ConfigDedicated_Default_TDD;
    }
    else{
      v_SoundingRS_UL_ConfigDedicated := cs_508_SoundingRsUl_ConfigDedicated_Default_FDD;
    }
    v_DedicatedSCell_UL_Configuration := cs_RadioResourceConfigDedicatedSCell_UL_Configuration_DEFAULT(-, -, v_SoundingRS_UL_ConfigDedicated, {srs_AntennaPort_r10 := an1});
    v_RRCConnectionReconfiguration := cs_RRCConnectionReconfiguration_Common(tsc_RRC_TI_Def, -, -, -, -, 
      cs_RRCConnectionReconfiguration_SCell_AddMod_IEs(cs_SCellToAddModList_OneSCell(cs_SCellToMod_r10(1,
        cs_RadioResourceConfigDedicatedSCell_r10(cs_PhysicalConfigDedicatedSCell_r10_DEFAULT(v_DedicatedSCell_UL_Configuration)))),-));
    SRB.send(cas_SRB1_RrcPdu_REQ(p_PCellId,
                                 cs_TimingInfo_Now,
                                 v_RRCConnectionReconfiguration));
    SRB.receive(car_SRB1_RrcPdu_IND(p_PCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));
    v_SubFrameInfo :=  f_EUTRA_GetNextSendOccasion(p_PCellId, 100); // 100 ms in future
    f_EUTRA_StopULGrantTransmission_No_PUCCH_Sync (p_PCellId,cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number) );
    //@siclog "Step 18" siclog@
    //Send TA for PTAG
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU(p_PCellId,tsc_RbId_DRB1, cs_TimingInfo(v_SubFrameInfo.SFN.Number, v_SubFrameInfo.Subframe.Number), cs_MAC_PDUList_1 (cs_MAC_PDU_TimingAdvance_0)));
    //@siclog "Step 19" siclog@
    // Send TA for Stag1
    v_SubFrameInfo2 :=  f_EUTRA_TimingInfoAdd(v_SubFrameInfo, 300); // 100 ms in future
    DRB.send (cas_DRB_COMMOM_REQ_MAC_PDU(p_SCellId,tsc_RbId_DRB1, cs_TimingInfo(v_SubFrameInfo2.SFN.Number, v_SubFrameInfo2.Subframe.Number), cs_MAC_PDUList_1 (cs_MAC_PDU_TimingAdvance_0_STAG1)));
    //@siclog "Step 20" siclog@
    // Send DL PDU on CC1
    v_SubFrameInfo2 :=  f_EUTRA_TimingInfoAdd(v_SubFrameInfo, 600); // 600 ms in future
    DRB.send(cas_DRB_COMMOM_REQ_MAC_PDU(p_SCellId,
                                        tsc_RbId_DRB1,
                                        cs_TimingInfo(v_SubFrameInfo2.SFN.Number, v_SubFrameInfo2.Subframe.Number),
                                        cs_MAC_PDUList_1(cs_MAC_PDU_NoPadding_RLCPDU_LCID(tsc_LchId_DRB1, v_EncodedRlcPdu3))));
    //Check UE stops SRS in parallel to step 22
   //@siclog "Step 21-22" siclog@
   interleave {
       []SYSIND.receive ( car_PRACH_Preamble_IND_Any(p_PCellId, cr_TimingInfo_Any)) -> value v_SYSTEM_IND
       {
            v_SubFrameTimingRX := v_SYSTEM_IND.Common.TimingInfo.SubFrame;
             if (f_EUTRA_SubFrameTimingDuration_Float(v_SubFrameInfo2, v_SubFrameTimingRX) <= 0.225)
               {
                     f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 22");
                  }
             else
                 {
                       //f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 22");
                       v_SetVerdictFalse := true;
                 }
       }
     [] SYSIND.receive(car_SRS_IND(p_SCellId, cr_TimingInfo_Any, tsc_SRS_Stop)) {}
   };
   // defined to avoid usage of communication operation not allowed inside an interleave statement
   if (v_SetVerdictFalse) {
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 22");
   }
   //@siclog "Step 23" siclog@
   //Send an uplink grant
   f_EUTRA_OneULGrantTransmission( p_PCellId,
                                    cs_TimingInfo_Now,
                                    7, 3);  //results in a grant of 328 bits= 41 B
   //@siclog "Step 24" siclog@
   DRB.receive (car_DRB_COMMON_IND_RLC_PDUList (p_PCellId,
                                                 tsc_RbId_DRB1,
                                                 cr_TimingInfo_Any,
                                                 {cr_RLC_AMD_1PDU(?, int2bit(2, 10), v_EncodedPdcpPdu3)}));
    // config for stop of PReamble reception indication in Scell
    f_EUTRA_SS_ConfigRachPreambleIndMode(p_SCellId, tsc_L1Mac_IndicationMode_Disable);
    // config for stop of PReamble reception indication in Pcell
    f_EUTRA_SS_ConfigRachPreambleIndMode(p_PCellId, tsc_L1Mac_IndicationMode_Disable);
    // Config for stop of SRS toggle indication
    f_SS_ConfigSRS_IndMode (p_SCellId, tsc_L1Mac_IndicationMode_Disable );
    // deactivate reception of SRS's in alt step
    deactivate(v_DefaultVar);
    //@siclog "Step 25" siclog@
    //Send RLC status PDU on CC2
    f_RLC_Status_PDU_Tx(tsc_LchId_DRB1,3,p_PCellId);
    ///////////////////////
    // Postamble
    ///////////////////////
    f_EUTRA_StartULGrantTransmissionOnSR_Reception_No_PUCCH_Sync(p_PCellId, cs_TimingInfo_Now, cs_DciInfo_CcchDcchDtchUL_Def);
    f_EUTRA_TestBody_Set(false);
    //Switch/power off UE
    f_EUTRA_Postamble(p_PCellId, E2_CONNECTED);
  }
  /*
   * @desc      Common function : REFERENCE TS 36.523-1 clause 7.1.2.11.1
   * @status
   */
  function f_TC_7_1_2_11_1_EUTRA() runs on EUTRA_PTC
  { /*CA / Maintenance of uplink time alignment / Multiple TA/ Intra-band Contiguous CA */
    f_TC_7_1_2_11_Common(eutra_Cell1, eutra_Cell3,CA_IntraBand);
  }
  /*
   * @desc      Common function : REFERENCE TS 36.523-1 clause 7.1.2.11.2
   * @status
   */
  function f_TC_7_1_2_11_2_EUTRA() runs on EUTRA_PTC
  { /* CA / Maintenance of uplink time alignment / Multiple TA /  Inter-Band CA */
    f_TC_7_1_2_11_Common(eutra_Cell1, eutra_Cell10, CA_InterBand);
  }
  /*
   * @desc      Common function : REFERENCE TS 36.523-1 clause 7.1.2.11.3
   * @status
   */
  function f_TC_7_1_2_11_3_EUTRA() runs on EUTRA_PTC
  { /*CA / Maintenance of uplink time alignment / Multiple TA/ Intra-band Non Contiguous CA */
    f_TC_7_1_2_11_Common(eutra_Cell1, eutra_Cell3, CA_IntraBandNonContiguous);//@sic R5s150322 sic@
  }
}
... 
·   Modification in module ‘EUTRA_CommonFunctions_CA.ttcn’
	Function name
	f_EUTRA_508RRC_AddMod_1Scell ()

	Reason for change
	See change 2.

	Summary of change
	See change 2.

	TTCN module
	Common\EUTRA_CA\EUTRA_CommonFunctions_CA.ttcn

	MCC160 Comment
	Accepted


Before change

...
  /*
   * @desc      Generic function to add or modify a Scell
   * @param     p_PCellId
   * @param     p_SCellId
   * @param     p_SCellIndex
   * @param     p_CA_Type
   * @param     p_UL_CA             (default value: false)
   * @param     p_C_RNTI            (default value: tsc_C_RNTI_Def)
   * @param     p_MeasGapConfig     (default value: omit)
   * @param     p_RRCConnectionReconfiguration (default value: omit)
   * @param     p_DciUlInfo_SCell   (default value: cs_DciInfo_CcchDcchDtchUL_Def)
   * @param     p_UL_GrantConfig_SCell (default value: cs_UL_GrantConfig_Def)
   * @param     p_STAG_ToReleaseList (default value: omit)
   * @param     p_STAG_ToAddModList (default value: omit)
   * @param     p_MAC_MainConfigSCell_r11 (default value: omit)
   * @param     p_RACH_ConfigCommonSCell_r11 (default value: omit)
   * @param     p_PRACH_Config_SCell (default value: omit)
   * @param     p_TimeAlignmentTimer (default value: sf750)
   * @param     p_ScellDeactivationTimer (default value: rf32)
   * @param     p_MBSFN_SubframeConfigList (default value: omit)
   * @status    APPROVED (LTE_A_R10_R11)
   */
  function f_EUTRA_508RRC_AddMod_1Scell(EUTRA_CellId_Type p_PCellId,
                                        EUTRA_CellId_Type p_SCellId,
                                        SCellIndex_r10 p_SCellIndex,
                                        CA_Tested_Type p_CA_Type,
                                        boolean p_UL_CA := false, // UL CA desired
                                        C_RNTI p_C_RNTI := tsc_C_RNTI_Def,
                                        template (omit) MeasGapConfig p_MeasGapConfig := omit, //@sic R5s130503 sic@
                                        template (omit) DL_DCCH_Message p_RRCConnectionReconfiguration := omit,
                                        template (omit)  DciUlInfo_Type p_DciUlInfo_SCell := cs_DciInfo_CcchDcchDtchUL_Def,
                                        template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SCell := cs_UL_GrantConfig_Def,
                                        //@sic new CA-Enh test development sic@
                                        template (omit) STAG_ToReleaseList_r11 p_STAG_ToReleaseList := omit,
                                        template (omit) STAG_ToAddModList_r11 p_STAG_ToAddModList := omit,
                                        template (omit)  MAC_MainConfigSCell_r11  p_MAC_MainConfigSCell_r11 := omit,
                                        template (omit) RACH_ConfigCommonSCell_r11 p_RACH_ConfigCommonSCell_r11 := omit,
                                        template (omit) PRACH_Config  p_PRACH_Config_SCell := omit,
                                        TimeAlignmentTimer p_TimeAlignmentTimer := sf750,
                                        MAC_MainConfig_ScellDeactivationTimer_Type p_ScellDeactivationTimer := rf32, //@sic R5s150285 sic@
                                        template(omit) MBSFN_SubframeConfigList p_MBSFN_SubframeConfigList := omit, //@sic R5s160162 sic@
                                        template (omit) PHR_Config_Type p_PHR_Config := cs_PHR_Config_Setup(sf500, sf200, dB3))
    runs on EUTRA_PTC
  {
    var integer v_Arfcn_SCell := f_EUTRA_CellInfo_GetEARFCN_DL(p_SCellId);
    var PhysCellId v_PhysicalCellIdentity_Scell := f_EUTRA_CellInfo_GetPhyCellId(p_SCellId);
    var SCell_DL_Bandwidth_Type v_SCell_DL_Bandwidth := f_EUTRA_CellInfo_GetSCell_DL_ChBandwidth(p_SCellId);
    var template (value) RachProcedureConfig_Type v_RachProcedureConfig;
    var template (value) DL_DCCH_Message v_RRCConnectionReconfiguration;
    var EUTRA_FDD_TDD_CellInfo_Type v_FDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info (p_SCellId);
    var template(omit) TDD_Config v_TDD_Config := omit;  // omitted in FDD
    var boolean v_UL_CA := p_UL_CA;
    var template (omit) CommonSCell_UL_Configuration_Type v_CommonSCell_UL_Configuration := omit;
    var template (omit)  DedicatedSCell_UL_Configuration_Type v_DedicatedSCell_UL_Configuration := omit;
    var template (value) SCellToAddMod_r10 v_SCellToAddMod;
    var Ul_Bandwidth_Type v_UL_Bandwidth := f_EUTRA_CellInfo_GetUL_ChBandwidth(p_SCellId);
    var Dl_Bandwidth_Type v_DL_Bandwidth := f_EUTRA_CellInfo_GetDL_ChBandwidth(p_SCellId);
    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam(v_DL_Bandwidth, v_UL_Bandwidth);
    var EUTRA_CellSysInfo_Type v_EUTRA_CellSysInfo_Type := f_EUTRA_CellInfo_GetSYSINFO(p_SCellId);
    var RadioResourceConfigCommonSIB v_RadioResourceConfigCommonSIB := v_EUTRA_CellSysInfo_Type.BCCH_Info.SIs[0].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib2.radioResourceConfigCommon;
    var template (value) Scell_Capability_Type v_Scell_Capability := DlOnly;
    var template (value) PUCCH_ConfigDedicated_v1020 v_PUCCH_ConfigDedicated;
    if (v_FDD_TDDInfo.FDD_TDD == TDD) {  // initialise tdd-Configuration  to be used in
      v_TDD_Config := cs_TDD_ConfigDef(v_FDD_TDDInfo.TDD_SubframeAssignment);    // Value omitted in FDD
    }
    if (isvalue(p_RRCConnectionReconfiguration)) {
      v_RRCConnectionReconfiguration := valueof(p_RRCConnectionReconfiguration);
    } else {
      if (v_FDD_TDDInfo.FDD_TDD == TDD) {  // initialise tdd-Configuration  to be used in
        v_TDD_Config := cs_TDD_ConfigDef(v_FDD_TDDInfo.TDD_SubframeAssignment);    // Value omitted in FDD
        v_PUCCH_ConfigDedicated  := cs_508_PUCCH_ConfigDedicated_v1020_2TX(cs_N1PUCCH_AN_CS_r10_FDD_1TX, cs_N1PUCCH_AN_CS_r10_FDD_2TX); //@sic R5-133275 sic@
      }
      else { //FDD
        v_PUCCH_ConfigDedicated  := cs_508_PUCCH_ConfigDedicated_v1020_1TX(cs_N1PUCCH_AN_CS_r10_FDD_1TX);
        // Not addressed for 2TX
      }
      if (v_UL_CA) {
       //@sic R5-160646 sic@
        v_CommonSCell_UL_Configuration := cs_RadioResourceConfigCommonSCell_UL_Configuration_DEFAULT(v_ChannelBandwidthDependency.Pusch_HoppingOffset, v_RadioResourceConfigCommonSIB.soundingRS_UL_ConfigCommon);
        v_DedicatedSCell_UL_Configuration := cs_RadioResourceConfigDedicatedSCell_UL_Configuration_DEFAULT();//@sic R5s150070 sic@
      }
      v_SCellToAddMod := cs_SCellToAddMod_r10(p_SCellIndex,
                                              v_PhysicalCellIdentity_Scell,
                                              v_Arfcn_SCell,
                                              cs_508_RadioResourceConfigCommonSCell_r10_DEFAULT(v_SCell_DL_Bandwidth,
                                                                                                tsc_P_b_1Tx,
                                                                                                v_CommonSCell_UL_Configuration,
                                                                                                v_TDD_Config,
                                                                                                -,
                                                                                                p_RACH_ConfigCommonSCell_r11,
                                                                                                p_PRACH_Config_SCell,
                                                                                                p_MBSFN_SubframeConfigList), //@sic R5s150285 sic@
                                              cs_RadioResourceConfigDedicatedSCell_r10(cs_PhysicalConfigDedicatedSCell_r10_DEFAULT(v_DedicatedSCell_UL_Configuration),  p_MAC_MainConfigSCell_r11 ));//@sic new CA-Enh test development sic@
      v_RRCConnectionReconfiguration := cs_508_RRCConnectionReconfiguration_SCell_AddModRelease(tsc_RRC_TI_Def, v_SCellToAddMod, v_PUCCH_ConfigDedicated,-,p_TimeAlignmentTimer,-,-, p_STAG_ToReleaseList, p_STAG_ToAddModList,p_ScellDeactivationTimer,-,-,-, p_PHR_Config);//@sic new CA-Enh test development sic@
    }
    if (v_UL_CA) {
      v_Scell_Capability := UL_DL;
    }
    v_RachProcedureConfig := cs_RachProcedureConfig_CRNTI(f_EUTRA_CellInfo_GetRAR_TA(p_SCellId),
                                                          f_EUTRA_CellInfo_GetDL_ChBandwidth(p_SCellId));
    f_EUTRA_508RRC_AddModRel_Scell_Common(p_PCellId,
                                          {p_SCellId},
                                          -,
                                          v_RRCConnectionReconfiguration,
                                          v_Scell_Capability,
                                          v_RachProcedureConfig,
                                          p_C_RNTI,
                                          p_MeasGapConfig,
                                          p_DciUlInfo_SCell,
                                          p_UL_GrantConfig_SCell);
  }
...

After change

...
  
  /*
   * @desc      Generic function to add or modify a Scell
   * @param     p_PCellId
   * @param     p_SCellId
   * @param     p_SCellIndex
   * @param     p_CA_Type
   * @param     p_UL_CA             (default value: false)
   * @param     p_C_RNTI            (default value: tsc_C_RNTI_Def)
   * @param     p_MeasGapConfig     (default value: omit)
   * @param     p_RRCConnectionReconfiguration (default value: omit)
   * @param     p_DciUlInfo_SCell   (default value: cs_DciInfo_CcchDcchDtchUL_Def)
   * @param     p_UL_GrantConfig_SCell (default value: cs_UL_GrantConfig_Def)
   * @param     p_STAG_ToReleaseList (default value: omit)
   * @param     p_STAG_ToAddModList (default value: omit)
   * @param     p_MAC_MainConfigSCell_r11 (default value: omit)
   * @param     p_RACH_ConfigCommonSCell_r11 (default value: omit)
   * @param     p_PRACH_Config_SCell (default value: omit)
   * @param     p_TimeAlignmentTimer (default value: sf750)
   * @param     p_ScellDeactivationTimer (default value: rf32)
   * @param     p_MBSFN_SubframeConfigList (default value: omit)
   * @status    APPROVED (LTE_A_R10_R11)
   */
  function f_EUTRA_508RRC_AddMod_1Scell(EUTRA_CellId_Type p_PCellId,
                                        EUTRA_CellId_Type p_SCellId,
                                        SCellIndex_r10 p_SCellIndex,
                                        CA_Tested_Type p_CA_Type,
                                        boolean p_UL_CA := false, // UL CA desired
                                        C_RNTI p_C_RNTI := tsc_C_RNTI_Def,
                                        template (omit) MeasGapConfig p_MeasGapConfig := omit, //@sic R5s130503 sic@
                                        template (omit) DL_DCCH_Message p_RRCConnectionReconfiguration := omit,
                                        template (omit)  DciUlInfo_Type p_DciUlInfo_SCell := cs_DciInfo_CcchDcchDtchUL_Def,
                                        template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SCell := cs_UL_GrantConfig_Def,
                                        //@sic new CA-Enh test development sic@
                                        template (omit) STAG_ToReleaseList_r11 p_STAG_ToReleaseList := omit,
                                        template (omit) STAG_ToAddModList_r11 p_STAG_ToAddModList := omit,
                                        template (omit)  MAC_MainConfigSCell_r11  p_MAC_MainConfigSCell_r11 := omit,
                                        template (omit) RACH_ConfigCommonSCell_r11 p_RACH_ConfigCommonSCell_r11 := omit,
                                        template (omit) PRACH_Config  p_PRACH_Config_SCell := omit,
                                        TimeAlignmentTimer p_TimeAlignmentTimer := sf750,
                                        MAC_MainConfig_ScellDeactivationTimer_Type p_ScellDeactivationTimer := rf32, //@sic R5s150285 sic@
                                        template(omit) MBSFN_SubframeConfigList p_MBSFN_SubframeConfigList := omit, //@sic R5s160162 sic@
                                        template (omit) PHR_Config_Type p_PHR_Config := cs_PHR_Config_Setup(sf500, sf200, dB3),
                                        boolean p_setUlSrs := false )  runs on EUTRA_PTC
  {
    var integer v_Arfcn_SCell := f_EUTRA_CellInfo_GetEARFCN_DL(p_SCellId);
    var PhysCellId v_PhysicalCellIdentity_Scell := f_EUTRA_CellInfo_GetPhyCellId(p_SCellId);
    var SCell_DL_Bandwidth_Type v_SCell_DL_Bandwidth := f_EUTRA_CellInfo_GetSCell_DL_ChBandwidth(p_SCellId);
    var template (value) RachProcedureConfig_Type v_RachProcedureConfig;
    var template (value) DL_DCCH_Message v_RRCConnectionReconfiguration;
    var EUTRA_FDD_TDD_CellInfo_Type v_FDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info (p_SCellId);
    var template(omit) TDD_Config v_TDD_Config := omit;  // omitted in FDD
    var boolean v_UL_CA := p_UL_CA;
    var template (omit) CommonSCell_UL_Configuration_Type v_CommonSCell_UL_Configuration := omit;
    var template (omit)  DedicatedSCell_UL_Configuration_Type v_DedicatedSCell_UL_Configuration := omit;
    var template (value) SCellToAddMod_r10 v_SCellToAddMod;
    var Ul_Bandwidth_Type v_UL_Bandwidth := f_EUTRA_CellInfo_GetUL_ChBandwidth(p_SCellId);
    var Dl_Bandwidth_Type v_DL_Bandwidth := f_EUTRA_CellInfo_GetDL_ChBandwidth(p_SCellId);
    var ChannelBandwidthDependency_Type v_ChannelBandwidthDependency := f_EUTRA_BandDependentParam(v_DL_Bandwidth, v_UL_Bandwidth);
    var EUTRA_CellSysInfo_Type v_EUTRA_CellSysInfo_Type := f_EUTRA_CellInfo_GetSYSINFO(p_SCellId);
    var RadioResourceConfigCommonSIB v_RadioResourceConfigCommonSIB := v_EUTRA_CellSysInfo_Type.BCCH_Info.SIs[0].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib2.radioResourceConfigCommon;
    var template (value) Scell_Capability_Type v_Scell_Capability := DlOnly;
    var template (value) PUCCH_ConfigDedicated_v1020 v_PUCCH_ConfigDedicated;
    var template (value) SoundingRS_UL_ConfigDedicated v_SoundingRS_UL_ConfigDedicated := cs_508_SoundingRsUl_ConfigDedicated_Default_FDD;
    var template (value) SoundingRS_UL_ConfigDedicated_v1020 v_SoundingRS_UL_ConfigDedicated_v1020 := {srs_AntennaPort_r10 := an1};
    if (v_FDD_TDDInfo.FDD_TDD == TDD) {  // initialise tdd-Configuration  to be used in
      v_TDD_Config := cs_TDD_ConfigDef(v_FDD_TDDInfo.TDD_SubframeAssignment);    // Value omitted in FDD
      v_SoundingRS_UL_ConfigDedicated := cds_508_SoundingRsUl_ConfigDedicated_Default_TDD;
    }
    if (isvalue(p_RRCConnectionReconfiguration)) {
      v_RRCConnectionReconfiguration := valueof(p_RRCConnectionReconfiguration);
    } else {
      if (v_FDD_TDDInfo.FDD_TDD == TDD) {  // initialise tdd-Configuration  to be used in
        v_TDD_Config := cs_TDD_ConfigDef(v_FDD_TDDInfo.TDD_SubframeAssignment);    // Value omitted in FDD
        v_PUCCH_ConfigDedicated  := cs_508_PUCCH_ConfigDedicated_v1020_2TX(cs_N1PUCCH_AN_CS_r10_FDD_1TX, cs_N1PUCCH_AN_CS_r10_FDD_2TX); //@sic R5-133275 sic@
      }
      else { //FDD
        v_PUCCH_ConfigDedicated  := cs_508_PUCCH_ConfigDedicated_v1020_1TX(cs_N1PUCCH_AN_CS_r10_FDD_1TX);
        // Not addressed for 2TX
      }
      if (v_UL_CA) {
       //@sic R5-160646 sic@
                if (p_setUlSrs) {
          v_DedicatedSCell_UL_Configuration := cs_RadioResourceConfigDedicatedSCell_UL_Configuration_DEFAULT(-, cs_AntennaInfoUL_r10, v_SoundingRS_UL_ConfigDedicated, v_SoundingRS_UL_ConfigDedicated_v1020);
          f_EUTRA_SS_CommonCellConfig(p_SCellId, cas_CellConfig_ScellModSRS_AntennaInfo_REQ(p_SCellId, cs_TimingInfo_Now, f_EUTRA_CellInfo_GetSounding_ConfigCommon(p_SCellId), v_SoundingRS_UL_ConfigDedicated, v_SoundingRS_UL_ConfigDedicated_v1020));
        }
        else {
        v_CommonSCell_UL_Configuration := cs_RadioResourceConfigCommonSCell_UL_Configuration_DEFAULT(v_ChannelBandwidthDependency.Pusch_HoppingOffset, v_RadioResourceConfigCommonSIB.soundingRS_UL_ConfigCommon);
          v_DedicatedSCell_UL_Configuration := cs_RadioResourceConfigDedicatedSCell_UL_Configuration_DEFAULT();//@sic R5s150070 sic@
        }
      }
      v_SCellToAddMod := cs_SCellToAddMod_r10(p_SCellIndex,
                                              v_PhysicalCellIdentity_Scell,
                                              v_Arfcn_SCell,
                                              cs_508_RadioResourceConfigCommonSCell_r10_DEFAULT(v_SCell_DL_Bandwidth,
                                                                                                tsc_P_b_1Tx,
                                                                                                v_CommonSCell_UL_Configuration,
                                                                                                v_TDD_Config,
                                                                                                -,
                                                                                                p_RACH_ConfigCommonSCell_r11,
                                                                                                p_PRACH_Config_SCell,
                                                                                                p_MBSFN_SubframeConfigList), //@sic R5s150285 sic@
                                              cs_RadioResourceConfigDedicatedSCell_r10(cs_PhysicalConfigDedicatedSCell_r10_DEFAULT(v_DedicatedSCell_UL_Configuration),  p_MAC_MainConfigSCell_r11 ));//@sic new CA-Enh test development sic@
      v_RRCConnectionReconfiguration := cs_508_RRCConnectionReconfiguration_SCell_AddModRelease(tsc_RRC_TI_Def, v_SCellToAddMod, v_PUCCH_ConfigDedicated,-,p_TimeAlignmentTimer,-,-, p_STAG_ToReleaseList, p_STAG_ToAddModList,p_ScellDeactivationTimer,-,-,-, p_PHR_Config);//@sic new CA-Enh test development sic@
    }
    if (v_UL_CA) {
      v_Scell_Capability := UL_DL;
    }
    v_RachProcedureConfig := cs_RachProcedureConfig_CRNTI(f_EUTRA_CellInfo_GetRAR_TA(p_SCellId),
                                                          f_EUTRA_CellInfo_GetDL_ChBandwidth(p_SCellId));
    f_EUTRA_508RRC_AddModRel_Scell_Common(p_PCellId,
                                          {p_SCellId},
                                          -,
                                          v_RRCConnectionReconfiguration,
                                          v_Scell_Capability,
                                          v_RachProcedureConfig,
                                          p_C_RNTI,
                                          p_MeasGapConfig,
                                          p_DciUlInfo_SCell,
                                          p_UL_GrantConfig_SCell);
  }
... 
5. Execution Log Files

5.1 Qualcomm MSM 8996
The Qualcomm MSM 8996 UE passed this LTE-A CA 2UL test case on R&S CMW500 Multi-RAT Protocol Tester in CA_1A-5A band combination. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file
Rohde-Schwarz\tc_7_1_2_11_2_Qualcomm.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

